Cinemática de la particula: Triedro de Frenet

J.Mª Goicolea, 8/10/2000

Una particula se mueve sobre el paraboloide hiperbólico $z = \frac{x\,y}{2}$, estando su posición definida por

$x = 6\sin(k\,\xi)$, $y = -6\cos(k\,\xi)$.

Obtener los vectores del triedro intrínseco y el radio de curvatura, particularizando para $\xi = \frac{\pi}{3\,k}$.

```
> restart;
```

Definición de la trayectoria:

```
> x:=6*sin(k*xi);
> y:=-6*cos(k*xi);
> z:=1/2 * x*y;
```

$$x := 6\sin(k\,\xi)$$

$$y := -6\cos(k\,\xi)$$

$$z := -18\sin(k\,\xi)\cos(k\,\xi)$$

Escribimos el vector posición:

```
> with(linalg):
```

```
Warning, the protected names norm and trace have been redefined and unprotected
```

```
> vr:=vector([x,y,z]);
```

$$vr := [6\sin(k\,\xi), -6\cos(k\,\xi), -18\sin(k\,\xi)\cos(k\,\xi)]$$

Calculamos su derivada respecto del parametro:

```
> vrp:=map(diff,vr,xi);
```

$$vrp := [6\cos(k\,\xi)\,k, 6\sin(k\,\xi)\,k, -18\cos(k\,\xi)^2\,k + 18\sin(k\,\xi)^2\,k]$$

Definimos la función norma como raiz cuadrada del producto escalar de un vector por sí mismo

```
> norma:=proc(X)
sqrt(dotprod(X,X));
end:
```

Esta función sirve para calcular la derivada del arco respecto del parámetro ξ :

```
> sp:=simplify(norma(vrp),symbolic);
```

$$sp := 6\sqrt{2}\,k\sqrt{-18\cos(k\,\xi)^2 + 5 + 18\cos(k\,\xi)^4}$$

El vector tangente se obtiene normalizando la derivada de la curva:

```
> vt:=evalm(1/sp * vrp);
```

$$vt := \left[\frac{1}{2}\frac{\sqrt{2}\cos(k\,\xi)}{\sqrt{-18\cos(k\,\xi)^2 + 5 + 18\cos(k\,\xi)^4}}, \frac{1}{2}\frac{\sqrt{2}\sin(k\,\xi)}{\sqrt{-18\cos(k\,\xi)^2 + 5 + 18\cos(k\,\xi)^4}}, \frac{1}{12}\frac{\sqrt{2}\,(-18\cos(k\,\xi)^2 k + 18\,\mathrm{si}}{k\sqrt{-18\cos(k\,\xi)^2 + 5 + 18}}\right.$$

Particularizando en el punto pedido

```
> vt1:=subs(xi=Pi/(3*k),evalm(vt));
```

$$vt1 := \left[\frac{1}{2}\frac{\sqrt{2}\cos\left(\frac{1}{3}\pi\right)}{\sqrt{-18\cos\left(\frac{1}{3}\pi\right)^2 + 5 + 18\cos\left(\frac{1}{3}\pi\right)^4}}, \frac{1}{2}\frac{\sqrt{2}\sin\left(\frac{1}{3}\pi\right)}{\sqrt{-18\cos\left(\frac{1}{3}\pi\right)^2 + 5 + 18\cos\left(\frac{1}{3}\pi\right)^4}}, \frac{1}{12}\frac{\sqrt{2}\left(-18\cos\left(\frac{1}{3}\pi\right)^2 k + 18\sin\left(\frac{1}{3}\pi\right)^2 k\right)}{k\sqrt{-18\cos\left(\frac{1}{3}\pi\right)^2 + 5 + 18\cos\left(\frac{1}{3}\pi\right)^4}}\right]$$

```
> vt1:=map(simplify,vt1);
```

$$vt1 := \left[\frac{1}{13}\sqrt{13}, \frac{1}{13}\sqrt{13}\sqrt{3}, \frac{3}{13}\sqrt{13}\right]$$

```
> vt1f:=map(evalf,vt1);
```

$$vt1f := [.2773500981, .4803844616, .8320502946]$$

Derivamos ahora de nuevo el vector tangente, respecto del parámetro ξ :

> **vtp:=map(diff,vt,xi);**

$$vtp := \left[-\frac{1}{4}\frac{\sqrt{2}\cos(k\,\xi)\,(36\cos(k\,\xi)\sin(k\,\xi)\,k-72\cos(k\,\xi)^3\sin(k\,\xi)\,k)}{(-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4)^{\left(\frac{3}{2}\right)}} - \frac{1}{2}\frac{\sqrt{2}\sin(k\,\xi)\,k}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi}} \right.$$

$$-\frac{1}{4}\frac{\sqrt{2}\sin(k\,\xi)\,(36\cos(k\,\xi)\sin(k\,\xi)\,k-72\cos(k\,\xi)^3\sin(k\,\xi)\,k)}{(-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4)^{\left(\frac{3}{2}\right)}} + \frac{\frac{1}{2}\sqrt{2}\cos(k\,\xi)\,k}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}},$$

$$-\frac{1}{24}\frac{\sqrt{2}\,(-18\cos(k\,\xi)^2k+18\sin(k\,\xi)^2k)\,(36\cos(k\,\xi)\sin(k\,\xi)\,k-72\cos(k\,\xi)^3\sin(k\,\xi)\,k)}{k\,(-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4)^{\left(\frac{3}{2}\right)}}$$

$$\left. + \frac{6\sqrt{2}\,k\cos(k\,\xi)\sin(k\,\xi)}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}} \right]$$

> **vtp:=map(simplify,vtp,symbolic);**

$$vtp := \left[\frac{1}{2}\frac{\sqrt{2}\sin(k\,\xi)\,k\,(18\cos(k\,\xi)^4-5)}{(-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4)^{\left(\frac{3}{2}\right)}}, -\frac{1}{2}\frac{\sqrt{2}\cos(k\,\xi)\,k\,(13-36\cos(k\,\xi)^2+18\cos(k\,\xi)^4)}{(-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4)^{\left(\frac{3}{2}\right)}}, \right.$$

$$\left. 3\frac{\sqrt{2}\,k\cos(k\,\xi)\sin(k\,\xi)}{(-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4)^{\left(\frac{3}{2}\right)}} \right]$$

El vector normal principal se obtiene normalizando esta derivada:

> **vtpn:=simplify(norma(vtp),symbolic);**

$$vtpn := \frac{1}{2}\frac{k\sqrt{-108\cos(k\,\xi)^4+108\cos(k\,\xi)^2+10}}{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}$$

> **vn:=evalm((1/vtpn)*vtp);**

$$vn := \left[\frac{\sqrt{2}\sin(k\,\xi)\,(18\cos(k\,\xi)^4-5)}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}\,\sqrt{-108\cos(k\,\xi)^4+108\cos(k\,\xi)^2+10}}, -\frac{\sqrt{2}\cos(k\,\xi)\,(13-36\cos(k\,\xi)^2+18\cos(k\,\xi)^4)}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}\,\sqrt{-108\cos(k\,\xi)^4+108\cos(k\,\xi)^2+10}}, 6\frac{\sqrt{2}\cos(k\,\xi)\sin(k\,\xi)}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}\,\sqrt{-108\cos(k\,\xi)^4+108\cos(k\,\xi)^2+10}} \right]$$

> **vn:=map(simplify,vn,symbolic);**

$$vn := \left[\frac{\sin(k\,\xi)\,(18\cos(k\,\xi)^4-5)}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}\,\sqrt{-54\cos(k\,\xi)^4+54\cos(k\,\xi)^2+5}}, -\frac{\cos(k\,\xi)\,(13-36\cos(k\,\xi)^2+18\cos(k\,\xi)^4)}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}\,\sqrt{-54\cos(k\,\xi)^4+54\cos(k\,\xi)^2+5}}, 6\frac{\sin(k\,\xi)\cos(k\,\xi)}{\sqrt{-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4}\,\sqrt{-54\cos(k\,\xi)^4+54\cos(k\,\xi)^2+5}} \right]$$

> **vn1:=map2(subs,xi=Pi/(3*k),evalm(vn)):**

> **vn1:=map(simplify,vn1);**

$$vn1 := \left[-\frac{31}{286}\sqrt{13}\sqrt{3}, -\frac{41}{286}\sqrt{13}, \frac{12}{143}\sqrt{13}\sqrt{3} \right]$$

> **vn1f:=map(evalf,vn1);**

$$vn1f := [-.6769053778, -.5168797285, .5240557763]$$

Calculamos ahora el radio de curvatura, en la posición pedida, como el inverso del módulo de la derivada del vector tangente respecto del arco s . (Esta derivada se evalúa derivando primero la tangente respecto del parámetro ξ .)

> **R:=sp/vtpn;**

$$R := 12\,\frac{\sqrt{2}\,\left(-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4\right)^{\left(\frac{3}{2}\right)}}{\sqrt{-108\cos(k\,\xi)^4+108\cos(k\,\xi)^2+10}}$$

\> **R1:=simplify(eval(subs(xi=Pi/(3*k),R)));**

\> **R1f:=evalf(R1);**

$$R1 := \frac{39}{22}\sqrt{13}$$

$$R1f := 6.391659081$$

Otra forma de calcular la curvatura es mediante la siguiente fórmula:

\> **vrpp:=map(diff,vrp,xi);**

\> **kappa:=norma(crossprod(vrp,vrpp))/norma(vrp)^3;**

$$vrpp := [-6\sin(k\,\xi)\,k^2,\,6\cos(k\,\xi)\,k^2,\,72\cos(k\,\xi)\,k^2\sin(k\,\xi)]$$

$$\kappa := \mathrm{sqrt}\Big((324\sin(k\,\xi)^2 k^3\cos(k\,\xi)+108\cos(k\,\xi)^3 k^3)\,\overline{(324\sin(k\,\xi)^2 k^3\cos(k\,\xi)+108\cos(k\,\xi)^3 k^3)}$$
$$+\,(-324\cos(k\,\xi)^2 k^3\sin(k\,\xi)-108\sin(k\,\xi)^3 k^3)\,\overline{(-324\cos(k\,\xi)^2 k^3\sin(k\,\xi)-108\sin(k\,\xi)^3 k^3)}$$
$$+\,(36\cos(k\,\xi)^2 k^3+36\sin(k\,\xi)^2 k^3)\,\overline{(36\cos(k\,\xi)^2 k^3+36\sin(k\,\xi)^2 k^3)}\Big)\Big/\Big($$
$$36\cos(k\,\xi)\,k\,\overline{(\cos(k\,\xi)\,k)}+36\sin(k\,\xi)\,k\,\overline{(\sin(k\,\xi)\,k)}+(-18\cos(k\,\xi)^2 k+18\sin(k\,\xi)^2 k)\,\overline{(-18\cos(k\,\xi)^2 k+1}$$
$$^{\left(\frac{3}{2}\right)}$$
$$\Big)$$

\> **kappa:=simplify(kappa,symbolic);**

$$\kappa := \frac{1}{24}\,\frac{\sqrt{-108\cos(k\,\xi)^4+108\cos(k\,\xi)^2+10}\,\sqrt{2}}{\left(-18\cos(k\,\xi)^2+5+18\cos(k\,\xi)^4\right)^{\left(\frac{3}{2}\right)}}$$

Por ultimo, obtenemos la binormal

\> **vb:=simplify(crossprod(vt,vn),symbolic);**

$vb :=$

$$\left[-\frac{3}{2}\frac{(-3+2\cos(k\xi)^2)\cos(k\xi)\sqrt{2}}{\sqrt{-54\cos(k\xi)^4+54\cos(k\xi)^2+5}}, -\frac{3}{2}\frac{(2\cos(k\xi)^2+1)\sin(k\xi)\sqrt{2}}{\sqrt{-54\cos(k\xi)^4+54\cos(k\xi)^2+5}}, \frac{1}{2}\frac{\sqrt{2}}{\sqrt{-54\cos(k\xi)^4+54\ \mathrm{co}}}\right.$$

```
> vb1:=simplify(subs(xi=Pi/(3*k),evalm(vb)));
```

$$vb1 := \left[\frac{15}{22}, -\frac{9}{22}\sqrt{3}, \frac{2}{11}\right]$$

```
> vb1f:=map(evalf,vb1);
```

$$vb1f := [.6818181818, -.7085662397, .1818181818]$$

Dibujamos la curva junto con los vectores del triedro

```
> with(plots):with(plottools):
```

```
Warning, the name changecoords has been redefined
```

damos el valor unidad al parametro k

```
> c1:=subs(k=1,evalm(vr));
```

$$c1 := [6\sin(\xi), -6\cos(\xi), -18\sin(\xi)\cos(\xi)]$$

La función plots[spacecurve] crea el dibujo de la curva en 3D

```
> curva:=spacecurve(c1,xi=0..Pi,color=brown,thickness=3):
```

la función plottools[arrow] permite dibujar las flechas

```
> base:=subs(xi=Pi/3,evalm(c1)):
> tta:=arrow(base,evalm(3*vt1),[1,0,0],.3,.9,.1,color=green):
> nna:=arrow(base,evalm(3*vn1),[1,0,0],.3,.9,.1,color=red):
> bba:=arrow(base,evalm(3*vb1),[1,0,0],.3,.9,.1,color=black):
```

Dibujamos ahora todo el conjunto

```
> surf:=plot3d(u*v/2,u=0..6,v=-6..6):
> display(curva,tta,nna,bba,surf,view=-9..9,
axes=BOXED,scaling=CONSTRAINED,shading=Z,orientation=[30,75]);
```



Animación

> **c2:=convert(c1,list);**

$$c2 := [6\sin(\xi), -6\cos(\xi), -18\sin(\xi)\cos(\xi)]$$

> **t2:=subs(k=1,convert(evalm(3*vt),list)):**

> **n2:=subs(k=1,convert(evalm(3*vn),list)):**

> **b2:=subs(k=1,convert(evalm(3*vb),list)):**

> **c3:=unapply(c2,xi);**

> **t3:=unapply(t2,xi):**

> **n3:=unapply(n2,xi):**

> **b3:=unapply(b2,xi):**

$$c3 := \xi \to [6\sin(\xi), -6\cos(\xi), -18\sin(\xi)\cos(\xi)]$$

> **curva:=spacecurve(c2,xi=0..Pi,color=brown,thickness=3):**

```
> surf:=plot3d(u*v/2,u=0..6,v=-6..6,style=contour):

> nump:=30:
dp:=Pi/nump:

> Q:=seq(display(curva,surf,view=-9..9,
arrow(c3(dp*n),evalf(t3(dp*n)),
[1,0,0],.3,.9,.1,color=green),
arrow(c3(dp*n),evalf(n3(dp*n)),
[1,0,0],.3,.9,.1,color=red),
arrow(c3(dp*n),evalf(b3(dp*n)),
[1,0,0],.3,.9,.1),color=black),
n=0..nump):

> display(Q,insequence=true,
axes=none,scaling=CONSTRAINED,shading=Z,orientation=[30,75]);
```